


# The Revised Oberon-2 programming language

Andreas Pirklbauer

8.4.2019

*Revised Oberon-2* is a revision of the programming language Oberon-2<sup>1</sup>. The main difference to the original is that it is a superset of *Revised Oberon (Oberon-07)* as defined in 2007/2016<sup>2</sup> rather than being based on the original language *Oberon* as defined in 1988/1990<sup>3</sup>.



This document describes only the additions or changes made to *Revised Oberon (Oberon-07)*, such as type-bound procedures, a dynamic heap allocation procedure for fixed-length and open arrays, a numeric case statement, exporting and importing of string constants, and no access to intermediate objects within nested scopes.

## Type-bound procedures

Globally declared procedures may be associated with a record type declared in the same module. The procedures are said to be *bound* to the record type. The binding is expressed by the type of the *receiver* in the heading of a procedure declaration. The receiver may be either a variable parameter of record type *T* or a value parameter of type *POINTER TO T* (where *T* is a record type). The procedure is bound to the type *T* and is considered local to it.

ProcedureHeading = PROCEDURE [Receiver] IdentDef [FormalParameters].  
Receiver = "(" [VAR] ident ":" ident ")".

If a procedure *P* is bound to a type *T0*, it is implicitly also bound to any type *T1* which is an extension of *T0*. However, a procedure *P'* (with the same name as *P*) may be explicitly bound to *T1* in which case it overrides the binding of *P*. *P'* is considered a *redefinition* of *P* for *T1*. The formal parameters of *P* and *P'* must match. If *P* and *T1* are exported, *P'* must be exported too.

If *v* is a designator and *P* is a type-bound procedure, then *v.P* denotes that procedure *P* which is bound to the dynamic type of *v*. This may be a different procedure than the one bound to the static type of *v*. *v* is passed to *P*'s receiver according to the standard parameter passing rules.

If *r* is a receiver parameter declared with type *T*, *r.P*<sup>^</sup> (pronounced *r.P-referenced*) denotes the (redefined) procedure *P* bound to the base type of *T*.

Example:

<sup>1</sup> Mössenböck H., Wirth N.: *The Programming Language Oberon-2. Structured Programming*, 12(4):179-195, 1991

<sup>2</sup> <http://www.inf.ethz.ch/personal/wirth/Oberon/Oberon07.Report.pdf> (Revision 3.5, 2016)

<sup>3</sup> <https://inf.ethz.ch/personal/wirth/Oberon/Oberon.Report.pdf> (Revision 1.10, 1990)

```

1  MODULE Trees;
2  IMPORT Out;
3
4  TYPE Tree = POINTER TO Node;
5  Node = RECORD key : INTEGER;
6    left, right: Tree
7  END ;
8
9  CenterTree = POINTER TO CenterNode;
10 CenterNode = RECORD (Node) width: INTEGER;
11   subnode: Tree
12 END ;
13
14 PROCEDURE (T: Tree) Insert (node: Tree);          (*procedure bound to Tree*)
15   VAR p, father: Tree;
16 BEGIN p := T;
17   REPEAT father := p;
18     IF node.key < p.key THEN p := p.left
19     ELSIF node.key > p.key THEN p := p.right
20     ELSE p := NIL
21     END
22   UNTIL p = NIL;
23   IF node.key < father.key THEN father.left := node ELSE father.right := node END ;
24   node.left := NIL; node.right := NIL
25 END Insert;
26
27
28 PROCEDURE (T: CenterTree) Insert (node: Tree);   (*redefinition of Insert bound to CenterTree*)
29 BEGIN Out.Int(node(CenterTree).width, 3);
30   T.Insert^(node)                                (*calls the Insert procedure bound to Tree*)
31 END Insert;
32
33 END Trees.

```

## Dynamic heap allocation procedure for fixed-length and open arrays

If  $p$  is a variable of type  $P = \text{POINTER TO } T$ , a call of the predefined procedure *NEW* allocates a variable of type  $T$  in free storage at run time. The type  $T$  can be a record or array type.

If  $T$  is a record type or an array type with *fixed* length, the allocation has to be done with

NEW( $p$ )

If  $T$  is an *open* array type, the allocation has to be done with

NEW( $p$ ,  $len$ )

where  $T$  is allocated with the length given by the expression  $len$ , which must be an integer type.

In either case, a pointer to the allocated variable is assigned to  $p$ . This pointer  $p$  is of type  $P$ , while the referenced variable  $p^\wedge$  (pronounced *p-referenced*) is of type  $T$ .

If  $T$  is a record type, a field  $f$  of an allocated record  $p^\wedge$  can be accessed as  $p^\wedge.f$  or as  $p.f$ . If  $T$  is an array type, the elements of an allocated array  $p^\wedge$  can be accessed as  $p^\wedge[0]$  to  $p^\wedge[len-1]$  or as  $p[0]$  to  $p[len-1]$ , i.e. record and array selectors imply dereferencing.

If  $T$  is an array type, its element type can be a *record*, *pointer*, *procedure* or a *basic* type (BYTE, BOOLEAN, CHAR, INTEGER, REAL, SET), but not an *array* type (no multi-dimensional arrays).

Example:

```
1  MODULE Test;
2  TYPE R = RECORD x, y: INTEGER END ;
3  A = ARRAY OF R;           (*open array*)
4  B = ARRAY 20 OF INTEGER;   (*fixed-length array*)
5  P = POINTER TO A;         (*pointer to open array*)
6  Q = POINTER TO B;         (*pointer to fixed-length array*)
7
8  VAR a: P; b: Q;
9
10 PROCEDURE New1*;
11 BEGIN NEW(a, 100); a[53].x := 1
12 END New1;
13
14 PROCEDURE New2*;
15 BEGIN NEW(b); b[3] := 2
16 END New2;
17 END Test.
```

The following rules and restrictions apply:

- Bounds checks on *fixed-length* arrays are performed at *compile* time.
- Bounds checks on *open* arrays are performed at *run* time.
- If  $P$  is of type  $P = \text{POINTER TO } T$ , the type  $T$  must be a *named* record or array type<sup>4</sup>.

Note that allocating dynamic arrays requires a modified version of the inner core module *Kernel*, which introduces a new *kind* of heap block – *array* block in addition to *record* block<sup>5</sup> – and a modified garbage collector to handle such blocks. Array blocks allocated using  $\text{NEW}(p, \text{len})$  or  $\text{NEW}(p)$  are collected by the garbage collector in the same way as regular record blocks<sup>6</sup>.

## Numeric case statement

The Oberon-07 language definition<sup>7</sup> allows *numeric* case statements, but the official compiler does not implement them<sup>8</sup>. The revised compiler brings the compiler in line with the language report, and now also allows *numeric* CASE statements (*CASE integer OF*, *CASE char OF*) in addition to *type* case statements (*CASE pointer OF*, *CASE record OF*).

Case statements specify the selection and execution of a statement sequence according to the value of an expression. First the case expression is evaluated, then the statement sequence is executed whose case label list contains the obtained value. If the case expression is of type INTEGER or CHAR, all labels must be integers or single-character strings, respectively.

CaseStatement	=	CASE expression OF case {" " case} [ELSE StatementSequence] END.
case	=	[CaseLabelList ":" StatementSequence].
CaseLabelList	=	LabelRange {" " LabelRange}.
LabelRange	=	label [".." label].
label	=	integer   string   qualident.

<sup>4</sup> Restricting pointers to *named* arrays is consistent with the official Oberon-07 compiler, which restricts pointers to point to *named* (but not anonymous) records.

<sup>5</sup> In some implementations of the Oberon system, an additional kind of heap block describing a storage block of  $n$  bytes ("sysblk") exists in addition to array and record blocks, typically allocated by a special low-level procedure  $\text{SYSTEM.NEW}(p, n)$ . In our implementation, no such procedure is needed, as it is covered by a call to  $\text{NEW}(p, n)$ , where  $p$  is a pointer to an array of BYTE.

<sup>6</sup> The implementation of garbage collection on fixed-length and open arrays is similar to implementations in earlier versions of the Original Oberon system. See, for example, "Oberon

Technical Notes: Garbage collection on open arrays", J. Templ, ETH technical report, March 1991.

<sup>7</sup> <http://www.inf.ethz.ch/personal/wirth/Oberon/Oberon07.Report.pdf> (as of 3.5.2016)

<sup>8</sup> <http://www.projectoberon.com>

Example (*numeric* case statement):

```
1 CASE ch OF
2   "A" .. "Z": ReadIdentifier
3   | "0" .. "9": ReadNumber
4   | "'", '"': ReadString
5 ELSE SpecialCharacter
6 END
```

The type  $T$  of the case expression (case variable) may also be a record or pointer type. Then the case labels must be extensions of  $T$ , and in the statements  $S_i$  labelled by  $T_i$ , the case variable is considered as of type  $T_i$ . Case expressions of *type* case statements must be *simple* identifiers that cannot be followed by selectors, i.e. they cannot be elements of a structure (array elements or record fields).

Example (*type* case case statement):

```
1 TYPE R = RECORD a: INTEGER END ;
2 R0 = RECORD (R) b: INTEGER END ;
3 R1 = RECORD (R) b: REAL END ;
4 R2 = RECORD (R) b: SET END ;
5
6 P = POINTER TO R;
7 P0 = POINTER TO R0;
8 P1 = POINTER TO R1;
9 P2 = POINTER TO R2;
10
11 VAR p: P;
12
13 CASE p OF
14   | P0: p.b := 10
15   | P1: p.b := 2.5
16   | P2: p.b := {0, 2}
17 END
```

The following rules and restrictions apply:

- Case labels of numeric statements must have values between 0 and 255.
- If the value of the case expression does not correspond to any case label value in the source text, the statement sequence following the symbol ELSE is selected, if there is one. Otherwise the program is aborted (in the case of a *numeric* case statement)<sup>9</sup> or no action is taken (in the case of a *type* case statement).

Note that the ELSE clause has been re-introduced in the revised compiler for both the *numeric* and the *type* case statements, even though it is not part of the Oberon-07 language definition. This decision may be viewed as being somewhat in contradiction with conventional wisdom in programming language design, which suggests that the ELSE clause in a language construct should be reserved for the *exceptional* cases only, i.e. those that are neither numerous among the possible cases in the source text of a program nor do occur frequently at run time. If in addition the case label range is restricted (to values between 0 and 255 in our case), an ELSE clause should in most cases easily be expressible using explicit case label ranges.

---

<sup>9</sup> If one wants to treat such events as "empty" actions, an empty ELSE clause can be used.

The presence of an ELSE clause in the source text may also obfuscate the thinking of the programmer if, for example, program execution falls through to the ELSE clause unintentionally. In general, language constructs that allow program execution to continue or *fall through* to the next case or a “default” case are not recommended.

The use of an ELSE clause should be considered only in well-justified cases, for example if the index range (far) exceeds the label range. But even in that case, one should first try to find a representation using explicit case label ranges, as shown in the example below.

CASE i OF		CASE i OF
1: S1		1: S1
3: S3		3: S3
7: S7	is the same as	7: S7
9: S9		9: S9
ELSE S0		0, 2, 4-6, 8, 10-255: S0 (*preferred*)
END		END

## Exporting and importing of string constants

The Oberon-07 language definition allows exporting and importing of string constants, but the official compiler does not support it. The revised compiler implements this feature<sup>10</sup>.

Example:

```
1  MODULE M;
2    CONST string* = "This is a sample string";      (*exported string constant*)
3  END M.
4
5  MODULE N;
6    IMPORT M, Out;
7
8    PROCEDURE P*;
9      BEGIN Out.Str(M.string)                      (*print the imported string constant*)
10    END P;
11  END N.
```

Exported *string constants* are treated like (pre-initialized) exported *variables*. The symbol file contains the string's *export number* and *length*, but not the string itself. The object file contains the actual string, together with its *location* in the exporting module's area for *data*.

## No access to intermediate objects within nested scopes

The Oberon-07 language definition disallows access to *all* intermediate objects within nested scopes, but the official compiler enforces this only for intermediate *variables*. The revised compiler brings the compiler in line with the language report, and now also disallows access to intermediate *constants* and *types* in nested scopes, not just access to intermediate *variables*<sup>11</sup>.

Like the official Oberon-07 compiler, the revised compiler implements *shadowing through scope* when accessing named objects. This means when two objects share the same name, the one declared at the narrower scope hides, or shadows, the one declared at the wider scope. In such a situation, the *shadowed* element is not available in the narrower scope. If the *shadowing*

<sup>10</sup> <http://github.com/andreaspirklbauer/Oberon-importing-string-constants>

<sup>11</sup> <http://github.com/andreaspirklbauer/Oberon-no-access-to-intermediate-objects>

element is itself declared at an intermediate scope, it is only available at *that* scope level, but *not* in narrower scopes (as access to intermediate objects is disallowed).

The official Oberon-07 compiler already issues an error message, if intermediate *variables* are accessed within nested scopes (line 25 of the program below), *regardless* of whether a global variable with the same name exists (line 7) or not. With the revised compiler, the same error message is now *also* issued for intermediate *constants* (line 21) and *types* (lines 16 and 18).

Example:

```

1  MODULE Test;
2  CONST C = 10;           (*global constant C, shadowed in Q and therefore not available in R*)
3
4  TYPE G = REAL;         (*global type G, not shadowed in Q and therefore available in R*)
5  T = REAL;              (*global type T, shadowed in Q and therefore not available in R*)
6  VAR A,                  (*global variable A, not shadowed in Q and therefore available in R*)
7  B: INTEGER;            (*global variable B, shadowed in Q and therefore not available in R*)
8
9  PROCEDURE P;           (*global procedure P*)
10
11  PROCEDURE Q;           (*intermediate procedure Q, contains shadowing elements C, T and B*)
12  CONST C = 20;          (*intermediate constant C which shadows the global constant C*)
13  TYPE T = INTEGER;      (*intermediate type T which shadows the global type T*)
14  VAR B: INTEGER;        (*intermediate variable B which shadows the global variable B*)
15
16  PROCEDURE R(x: T): T;   (*access to intermediate type T allowed in original, not allowed in modified compiler*)
17  VAR i: INTEGER;
18  q: T;                  (*access to intermediate type T allowed in original, not allowed in modified compiler*)
19  g: G;                  (*access to global type G (not shadowed) allowed in both compilers*)
20  BEGIN (*R*)
21  i := C;                 (*access to interm. constant C allowed in original, not allowed in modified compiler*)
22  P;                     (*access to global (unshadowed) procedure P allowed in both compilers*)
23  Q;                     (*access to intermediate procedure Q allowed in both compilers*)
24  i := A;                 (*access to global (unshadowed) variable A allowed in both compilers*)
25  i := B;                 (*access to intermediate variable B not allowed in both compilers*)
26  RETURN i
27  END R;
28  END Q;
29  END P;
30  END Test.
```

Disallowing access to intermediate objects from within nested scopes while at the same time implementing *shadowing through scope* raises the question whether one should *relax* the shadowing rules somewhat and *allow* access to the *global* scope level, if an object with the same name as a global object is re-declared at an *intermediate* level, but *not* at the strictly local level ("piercing through the shadow").

In the above example, such an approach would allow access to the global variable *B* (line 7) in procedure *R* (line 25), effectively ignoring any variables *B* that may exist at an intermediate level (line 14). It would make nested procedures self-contained in the sense that they can be moved around freely. For example, procedure *R* can be moved from another place and be made local to procedure *Q* *without* having to worry about whether one can still access the global variable *B*. We have opted not to adopt this approach, in order not to break with the language tradition<sup>12</sup>.

<sup>12</sup> In the appendix of <http://github.com/andreaspirk/bauer/Oberon-no-access-to-intermediate-objects>, a possible implementation of such relaxed shadowing rules is provided.

## Appendix A: Additional features implemented by the Experimental Oberon compiler

A compiler for *Revised Oberon-2* has been implemented for the Experimental Oberon system<sup>13</sup>, a revision of the FPGA Oberon system<sup>14</sup>. Note that this compiler also implements a number of additional features, which are *not* part of the official *Revised Oberon-2* language definition, such as forward references and forward declarations of procedures and module contexts.

### Forward references of procedures

If a procedure Q which is local to another procedure P refers to the enclosing procedure P, as in

```
1  PROCEDURE P;  
2  
3  PROCEDURE Q;  
4    BEGIN (*body of Q*) P          (*forward reference from Q to P, as the body of P is not compiled yet *)  
5    END Q;  
6  
7  BEGIN (*body of P*) ...  
8  END P;
```

then the official Oberon-07 compiler generates the following code (on FPGA Oberon):

```
20  P'   BL 10          ... forward branch to line 31 (across the body of Q to the body of P)  
21  Q    body of Q  
...  
31  P    body of P      ... any calls from Q to P are BACKWARD jumps to line 20 and from there forward to line 31
```

whereas the revised compiler generates the following, more streamlined, code:

```
20  Q    body of Q  
...  
30  P    body of P      ... any calls from Q to P are FORWARD jumps to line 30, fixed up when P is compiled
```

i.e. it does not generate an extra forward jump in line 20 across the body of Q to the body of P and backward jumps from Q to line 20. With the official compiler, the extra BL instruction in line 20 is generated, so that Q can call P (as the body of Q is compiled before the body of P).

The main motivation for implementing *forward* references was *not* to improve the performance of nested procedure calls (the performance gain is negligible), but to allow for a straightforward implementation of forward *declarations* of procedures (see below). As a welcome side effect, generating forward calls for nested procedures actually makes the compiler (a little) *shorter*<sup>15</sup>.

### Forward declarations of procedures

Forward declarations of procedures have been eliminated in Oberon-07, as they can always be eliminated from any program by an appropriate nesting or by introducing procedure variables<sup>16</sup>.

Whether forward declarations of procedures *should* be re-introduced into the language can of course be debated. Here, we have experimented with them for 3 main reasons:

<sup>13</sup> <http://www.github.com/andreaspirklbauer/Oberon-experimental>

<sup>14</sup> <http://www.projectoberon.com>

<sup>15</sup> On the downside, it adds a hidden mechanism in the code generator to build up the fixup list of forward references (see procedures `ORG.load` and `ORG.Call`).

<sup>16</sup> See section 2 of [www.inf.ethz.ch/personal/wirth/Oberon/PortingOberon.pdf](http://www.inf.ethz.ch/personal/wirth/Oberon/PortingOberon.pdf)

- Legacy programs containing forward declarations of procedures are now accepted again by the compiler, without the need to re-express them using procedure variables.
- Introducing forward *declarations* of procedures adds only about a dozen of lines of source code to the compiler – once the mechanism of forward *references* is in place (see above). However, simplicity of implementation is generally not a good reason for introducing a given feature. Nevertheless, it is comforting to know that it *could* be done with minimal effort.
- Direct procedure calls are (slightly) more efficient than using procedure variables – although the performance gain is negligible and therefore provides no strong argument.

Forward declarations of procedures are implemented in exactly the same way as in the original Oberon implementation before the Oberon-07 language revision, i.e.,

- They are explicitly specified by `^` following the symbol `PROCEDURE` in the source text.
- The compiler processes the heading in the normal way, assuming its body to be missing. The newly generated object in the symbol table is marked as a forward declaration.
- When later in the source text the full declaration is encountered, the symbol table is first searched. If the given identifier is found and denotes a procedure, the full declaration is associated with the already existing entry in the symbol table and the parameter lists are compared. Otherwise a multiple definition of the same identifier is present.

In addition, in a forward declaration of a *type-bound procedure*, the receiver parameter must be of the same type as in the actual procedure declaration, and the formal parameter lists of both declarations must be identical.

Note that in our implementation *both* global and local procedures can be declared forward.

Example:

```

1  MODULE M;
2
3  PROCEDURE^ P(x, y: INTEGER; z: REAL);           (*forward declaration of P*)
4
5  PROCEDURE Q*;
6  BEGIN P(1, 2, 3.0)                               (*Q calls P which is declared forward*)
7  END Q;
8
9  PROCEDURE P(x, y: INTEGER; z: REAL);             (*procedure body of P*)
10 BEGIN ...
11 END P;
12 END M.
```

## Module contexts

*Module contexts* have originally been proposed for the A2 operating system<sup>17</sup>. A module context acts as a single-level name space for modules and allows modules with the same name to co-exist within different contexts.

```

Module   = MODULE ident [IN ident] “;”.
Import   = IMPORT ident [“:=” ident ] [IN ident] “;”.
```

<sup>17</sup> <http://www.ocp.inf.ethz.ch/wiki/Documentation/Language?action=download&upname=contexts.pdf>



In the first line, the optional identifier specifies the name of the context the module belongs to. In the second line, it tells the compiler in which context to look for when importing modules.

Module contexts are implemented as follows:

- Module contexts are specified within the *source* text of a module, as an optional feature. If a context is specified, the name of the source file typically (but not necessarily) contains a prefix indicating its module context, for example *Oberon.Texts.Mod* or *EO.Texts.Mod*.
- If a module context is specified within the source text of a module, the compiler generates the output files *contextname.modulename.smb* and *contextname.modulename.rsc*, i.e. the module context is encoded in the symbol and object file names.
- If no module context is specified within the source text of a module, the compiler generates the output files *modulename.smb* and *modulename.rsc*, i.e. no context is assumed.
- On an Experimental Oberon system, the module context "EO" is implicitly specified at run time, i.e. the module loader first looks for *EO.modulename.rsc*, then for *modulename.rsc*.
- A module cannot be loaded under more than one context on the same system.
- A module belonging to a context can only import modules belonging to the same context or no context (implementation restriction).

Example:

```
1  MODULE Test1 IN EO;           (*Experimental Oberon*)
2    IMPORT Out;
3
4    PROCEDURE Go1*;
5    BEGIN Out.Str("Hello from module Test1 in context EO (Experimental Oberon)"); Out.Ln
6    END Go1;
7
8  END Test1.
9
10
11  MODULE Test2 IN EO;          (*Experimental Oberon*)
12    IMPORT Out, Test1 IN EO;
13
14    PROCEDURE Go1*;
15    BEGIN Out.Str("Calling Test1.Go1.. "); Test1.Go1
16    END Go1;
17
18    PROCEDURE Go2*;
19    BEGIN Out.Str("Hello from module Test2 in context EO (Experimental Oberon)"); Out.Ln
20    END Go2;
21
22  END Test2.
```

## Appendix B: Implementation cost of the Revised Oberon-2 language additions

The implementation cost of the *Revised Oberon-2* language additions (including the additional experimental features implemented by the Experimental Oberon compiler) in source lines of code (sloc)<sup>18</sup> is as follows:

Compiler module	Revised Oberon (Oberon-07)	Revised Oberon-2	Difference	Percent
ORS (scanner)	293	293	0	0 %
ORB (base)	394	437	43	+ 10.9 %
ORG (generator)	984	1110	126	+ 12.8 %
ORP (parser)	949	1140	191	+ 20.1 %
<b>Total</b>	<b>2620</b>	<b>2982</b>	<b>360</b>	<b>+ 13.7 %</b>

Feature	Source lines of code
Type-bound procedures	220
Dynamic heap allocation procedure for fixed-length and open arrays	30
Numeric case statement	70
All other features combined	40
<b>Total</b>	<b>360</b>

\* \* \*

---

<sup>18</sup> Not counting about 100-150 additional lines of source code in modules *Kernel*, *Modules* and *System* to complement the implementation of the *Revised Oberon-2* language.