

The Experimental Oberon System

Andreas Pirklbauer

1.7.2017

Experimental Oberon¹ is a revision of the Original Oberon² operating system. It contains a number of enhancements including continuous fractional line scrolling with variable line spaces, multiple logical displays, enhanced viewer management, safe module unloading and a minimal version of the Oberon system building tools. Some of these modifications are purely of experimental nature, while others serve the explicit purpose of exploring potential future extensions, for example to add support for touch display devices.

1. Continuous fractional line scrolling with variable line spaces

Continuous fractional line scrolling has been added to the viewer system, enabling completely smooth scrolling of displayed texts with variable lines spaces and dragging of entire viewers with continuous content refresh. Both *far* (positional) scrolling and *near* (pixel-based) scrolling are realized³. To the purist, such a feature may represent an “unnecessary embellishment” of Oberon, but it is simply indispensable if the system is to support touch display devices where a mouse is absent and viewers may not have scrollbars. In such an environment, continuous scrolling is the only acceptable way to scroll and presents a more natural user interface. As a side effect, the initial learning curve for users new to the system is *considerably* reduced.

2. Multiple logical display areas (“virtual displays”)

The Oberon system was designed to operate on a *single* abstract logical display area which is decomposed into a number of vertical *tracks*, each of which is further decomposed into a number of horizontal *viewers*. Experimental Oberon adds the ability to create *several* such display areas (or *displays* for short) on the fly and to seamlessly switch between them. Thus, the extended conceptual hierarchy of the display system consists of the triplet (*display*, *track*, *viewer*), and consequently the underlying base module *Viewers* exports procedures to add and remove *displays*, to open and close *tracks* within displays and to open and close individual *viewers* within tracks. There are no restrictions on the number of displays, tracks or viewers that can be created. Focus viewers and text selections are separately defined for each display. This scheme naturally maps to systems with multiple *physical* monitors. It can also be used to realize fast context switching, for example in response to a swipe gesture on a touch display.

The command *System.OpenDisplay* opens a new logical display, *System.CloseDisplay* closes an existing one. *System.ShowDisplays* lists all open displays, *System.ThisDisplay* shows the display *id* and *name* of the current display, *System.SetDisplay* switches between displays, and *System.SetDisplayName* assigns a new name to an existing display.

¹ <http://www.github.com/andreaspirklbauer/Oberon-experimental>

² <http://www.inf.ethz.ch/personal/wirth/ProjectOberon/index.html> (Original Oberon, 2013 Edition); see also <http://www.projectoberon.com>

³ The system automatically switches back and forth between the two scrolling modes based on the horizontal position of the mouse pointer.

The command *System.Clone*, displayed in the title bar of every menu viewer, opens a new logical display on the fly *and* displays a copy of the initiating viewer *there*. The user can easily toggle between the two copies of the viewers (i.e. switch *displays*) with a single mouse click⁴.

Alternatively, the user can activate the command *System.Expand*, also displayed in the title bar of every menu viewer, to expand the viewer “as much as possible” by reducing all other viewers in the track to their minimum heights, and switch back to any of the “compressed” viewers by clicking on *System.Expand* again in any of their (still visible) title bars.

3. Enhanced viewer management

The basic viewer operations *Change* (in modules *Viewers* and *MenuViewers*) and *Modify* (in modules *MenuViewers* and *TextFrames*) have been generalized to handle *arbitrary* viewer modifications, including pure vertical translations (without changing the viewer’s height), adjusting the top bottom line, the bottom line and the height of a viewer with a single *viewer change* operation, and dragging multiple viewers around with a single *mouse drag* operation.

Several viewer message types (e.g., *ModifyMsg*) and identifiers (e.g., *extend*, *reduce*) have been eliminated, further streamlining the overall type hierarchy. The remaining message types and identifiers now have a single, well-defined purpose. For example, restoring a viewer is accomplished exclusively by means of a *restore* message identifier.

In addition, viewer message types that appeared to be generic enough to be made generally available to *all* viewer types have been merged and moved from higher-level modules to the base module *Viewers*, resulting in fewer module dependencies in the process. Most notably, module *TextFrames* no longer depends on module *MenuViewers*, making it now possible to (recursively) embed text frames into *other* types of frames or composite viewers, for example a viewer consisting of an *arbitrary* number of text, graphic or picture frames.

4. Safe module unloading

The semantics of module *unloading* has been refined as follows. If clients exist, a module or module group is never unloaded. If no clients *and* no references to a module or module group exist in the remaining modules or data structures, it is unloaded and its associated memory is released. If no clients, but references exist, the module or module group is *initially* removed only from the *list* of loaded modules, without releasing its associated memory⁵. Such *hidden* modules are later *physically* removed from *memory* as soon as there are no more references to them⁶. To achieve this automatic removal of module data, a new command *Modules.Collect*, which checks all possible combinations of module *groups*, has been added to the Oberon background task handling garbage collection⁷. Thus, module data is kept in memory as long as needed and removed from it as soon as possible. If a module *group* is to be unloaded and there exist clients or references *only* within this group, the group is unloaded *as a whole*.

⁴ By comparison, the Original Oberon commands *System.Copy* and *System.Grow* create a copy of the original viewer in the *same* (and only) logical display area – *System.Copy* opens another viewer in the same track of the display, *System.Grow* extends the viewer’s copy over the entire column or display, lifting the viewer to an “overlay” in the third dimension.

⁵ Removing a module from the list of loaded modules amounts to renaming it, allowing another module with the same name to be (re-)loaded again. Modules removed *only* from the list of loaded modules, but not from memory, are marked with an asterisk in the output of the command *System.ShowModules*. Commands of such “hidden” modules can be accessed by either specifying their module number or their (modified) module name, both of which are displayed by the command *System.ShowModules*. In both cases, the corresponding command text must be enclosed in double quotes. If a module *M* carries module number 14, for instance, one can activate a command *M.P* also by clicking on the text “14.P”. Typical use cases include hidden modules that still have Oberon background tasks installed which can only be removed by activating a command of the hidden modules themselves, or hidden modules that still have open viewers. If the command to close a viewer is displayed in the viewer’s menu bar, the user can manually edit the command text (by clicking within its bottom two pixel lines), replace the module name by its module number and enclose the modified command text in double quotes. Although this is somewhat clumsy, it at least enables the user to close the viewer. An alternative approach is to provide a “Close” command that also accepts the marked viewer as argument (using procedure *Oberon.MarkedViewer*). In this case, the command (with the module number used instead of the module name) can be activated from anywhere in the system, after first designating the viewer to be closed as operand by placing the “star” marker in it.

⁶ Removing a module from memory frees up the memory area previously occupied by the module block. In Original Oberon 2013 on RISC, this includes the module’s type descriptors. In some Oberon implementations, such as Original Oberon on Ceres, type descriptors are not placed in the module block, but allocated dynamically in the heap at module load time, in order to persist them beyond the lifetime of their associated modules. In Experimental Oberon, no such special precaution is necessary, because module blocks are removed only from the *list* of loaded modules, if they are still referenced by other loaded modules. Thus, type descriptors can safely be placed in the modules’ (static) module blocks.

⁷ The command *Modules.Collect* can be manually activated at any time. The command *System.Collect* also includes a call to procedure *Modules.Collect*.

The command *System.Free* attempts to unload the specified modules. It is also possible to unload *entire subsystems* of modules. Two typical cases are supported: unloading *imports* and *clients*. The command *System.FreeImports* attempts to unload the modules *and* all their direct and indirect *imports*, while *System.FreeRemovableImports* unloads only those modules that *can* be removed. The commands *System.FreeClients* and *System.FreeRemovableClients* are analogous. Note that these unloading strategies amount to heuristics and may tend to unload rather large parts of the system, which may not be desired. Therefore, the recommended way to unload modules is to use *System.Free* with a *specific* set of modules provided. To help identify valid subsets of modules, the commands *System.ShowImports*, *System.ShowClients*, *System.ShowRemovableImports* and *System.ShowRemovableClients* are provided.

References to a module can be either in the form of *type tags* (addresses of type descriptors) in *dynamic* objects of other modules (allocated via the predefined procedure NEW) pointing to descriptors of types declared in the module, or in the form of *procedure variables* installed in *static* or *dynamic* objects of other modules referring to procedures declared in the module⁸.

In sum, module unloading does not affect *past* references. For example, older versions of a module's code can still be executed if they are referenced by procedure variables in other loaded modules, even if a newer version of the module has been reloaded in the meantime. Type descriptors also remain accessible to other modules for as long as needed⁹.

Clients and references are checked *prior* to module unloading by selecting the modules to be unloaded using *Modules.Select*, and then invoking procedure *Modules.Check*¹⁰.

```

PROCEDURE Check*(VAR res: INTEGER); (*outside clients or references to selected modules*)
1  VAR mod: Module; pref, pvadr, r: LONGINT; resType, resProc: INTEGER; continue: BOOLEAN;
2  BEGIN CheckClients(res);
3  IF res = 0 THEN mod := root;
4  WHILE mod # NIL DO (*mark dynamic records reachable by all other modules*)
5    IF (mod.name[0] # 0X) & ~mod.selected THEN Kernel.Mark(mod.ptr) END ;
6    mod := mod.next
7  END ;
8  Kernel.Scan(Chk, Chk, resType, resProc); (*check dynamic type and procedure references*)
9  IF resType > 0 THEN res := 2 ELSIF resProc > 0 THEN res := 3
10 ELSE res := 0; mod := root; continue := TRUE;
11 WHILE continue & (mod # NIL) DO
12   IF (mod.name[0] # 0X) & ~mod.selected THEN
13     pref := mod.pvar; pvadr := Mem[pref];
14     WHILE continue & (pvadr # 0) DO r := Mem[pvadr];
15     IF Chk(r, continue) > 0 THEN res := 4 END ; (*check static procedure references*)
16     INC(pref, 4); pvadr := Mem[pref]
17   END
18   END ;
19   mod := mod.next
20 END
21 END
22 END
END Check;
```

⁸ An Oberon module can be viewed as a container of types, variables and procedures. Variables can be procedure-typed and can be declared as *global* variables (allocated in the module area when a module is loaded), as *local* variables (allocated on the stack when a procedure is called) or allocated as *dynamic* objects in the heap (via the predefined procedure NEW). Thus, in general there can be type, variable or procedure references from static or dynamic objects of other modules to static or dynamic objects of the modules to be unloaded. However, only *dynamic* type and *static* and *dynamic* procedure references need to be checked, for the following reasons. First, *static* type and variable references from other loaded modules referring (by name) to types or variables declared in the specified modules don't need to be checked, as these are handled via their import/export relationship (if clients exist, a module or module group is never unloaded). Second, *pointer* references from global or dynamic pointer variables of other loaded modules to *dynamic* objects (allocated in the heap) of the specified modules don't need to be checked, as such references are handled by the garbage collector during a future garbage collection cycle. And third, *pointer* references to *static* objects (declared as global variables) are only possible by resorting to low-level facilities and should be avoided (and, in fact, be disallowed).

⁹ If an older version of a module's code accesses global variables (of itself or of other modules), it will automatically access the "right" version of such variables – as it should.

¹⁰ Mem stands for the entire memory and assignments involving Mem are expressed as SYSTEM.GET(a, x) for x := Mem[a] and SYSTEM.PUT(a, x) for Mem[a] := x.

Clients are checked by verifying whether the number of clients of each module *within* the group of selected modules matches its *total* number of clients. References from *dynamic* objects in the heap are checked using a *mark-scan* scheme. During the *mark* phase (lines 3-7), objects reachable by *named* global pointer variables of *all other* modules are marked (line 5), thereby excluding heap records reachable *only* by the specified modules themselves. This ensures that when a module or module group is referenced *only* by itself, it can still be unloaded. The *scan* phase (line 8), implemented as a separate procedure *Scan* located in module *Kernel*¹¹, scans the heap sequentially (element by element), unmarks *marked* objects and checks whether the *type tags* of the encountered (marked) records point to descriptors of types declared in the selected modules, or whether *procedure variables* in these records refer to procedures declared in those modules. The latter check is then also performed for all *static* procedure variables of all other loaded modules (lines 11-20).

In order to omit in module *Kernel* any reference to the list of loaded modules rooted in module *Modules*, procedure *Kernel.Scan* is expressed as a *generic* heap scan scheme.

```

PROCEDURE Scan*(type, proc: Handler; VAR resType, resProc: INTEGER);
1  VAR p, r, mark, tag, size, offadr, offset: LONGINT; continue: BOOLEAN;
2  BEGIN p := heapOrg; resType := 0; resProc := 0; continue := TRUE;
3  REPEAT mark := Mem[p+4];
4    IF mark < 0 THEN (*free*) size := Mem[p]
5    ELSE (*allocated*) tag := Mem[p]; size := Mem[tag];
6    IF mark > 0 THEN Mem[p+4] := 0; (*unmark*)
7    IF continue THEN
8      IF type # NIL THEN INC(resType, type(tag, continue)) END ; (*call type for type tag*)
9      IF continue & (proc # NIL) THEN offadr := tag - 4; offset := Mem[offadr];
10     WHILE continue & (offset # -1) DO r := Mem[p+8+offset];
11       INC(resProc, proc(r, continue)); (*call proc for each procedure variable*)
12     DEC(offadr, 4); offset := Mem[offadr]
13   END
14   END
15   END
16   END
17   END ;
18   INC(p, size)
19   UNTIL p >= heapLim
END Scan;
```

Instead of directly checking whether the encountered (marked) heap records contain *type* or *procedure* references to the selected modules, two parametric handler procedures *type* and *proc* are called for individual elements of each record. Procedure *type* is called with the *type tag* of the record as argument (line 8), and procedure *proc* is called for each procedure variable declared in the record with the *procedure variable* itself as argument (line 11). An additional parameter *continue* allows the handler procedures to indicate to the caller that they should no longer be called (lines 7, 9, 10)¹². The results of the handler calls are *separately* added up for each handler procedure and returned in the two parameters *resType* and *resProc*.

Procedure *Modules.Check* uses this generic scheme by passing a (private) procedure *Chk* as the handler, which merely checks whether the argument supplied by *Kernel.Scan* (either a type tag or a procedure variable) references *any* of the selected modules to be unloaded. We note that procedure *Modules.Check* is not only called when a module is unloaded by the end user,

¹¹ The original procedure *Kernel.Scan* (implementing the scan phase of the garbage collector) has been renamed to *Kernel.Collect*, in analogy to *Modules.Collect*.

¹² If *one* of the two handler procedures sets *continue* to *FALSE*, procedure *Kernel.Scan* will stop calling *both* of them. This somewhat artificial restriction could be lifted if needed.

but also by the Oberon background process that removes no longer referenced *hidden* module data from memory. Thus, it must be able to handle *both* visible and hidden modules.

In order to make the outlined validation pass possible, type descriptors for *dynamic* objects¹³ allocated in the heap and descriptors of *global* module data located in *static* module blocks have been extended with a list of *procedure variable offsets*, adopting an approach employed in one of the earlier implementations of the Oberon system¹⁴, whose run-time representation of a dynamic record and its associated type descriptor is shown in Figure 1.




Figure 1: Run-time representation of a dynamic record and its type descriptor in MacOberon

To avoid traversing (and skipping over) the list of *pointer field offsets* in the type descriptor for *each* record encountered in the heap during the *scan* phase¹⁵, Experimental Oberon uses a slightly different run-time representation, where *procedure variable offsets* are *prepended*, rather than *appended*, to the existing fields of each type descriptor, as shown in Figure 2¹⁶.




Figure 2: Run-time representation of a dynamic record and its type descriptor in Experimental Oberon

¹³ See chapter 8.2, page 109, of the book *Project Oberon* (2013 Edition) for a detailed description of an Oberon type descriptor, which contains certain information about dynamically allocated records that is shared by all objects of the same record type (such as its size, information about type extensions and the offsets of pointer fields within the record).

¹⁴ <http://e-collection.library.ethz.ch/eserv/eth:3269/eth-3269-01.pdf> (The Implementation of MacOberon, 1990)

¹⁵ The traversal could also be avoided in other ways (e.g., by storing the number of pointer field offsets at a fixed location inside the type descriptor), but we opted for a simpler solution.

¹⁶ This runtime representation might get into conflict with some implementations of the Oberon-2 programming language, which may also prepend additional run-time information to the fields in each type descriptor (typically a "method table" associated with each Oberon-2 type descriptor, which however serves a different purpose than the list of procedure variable offsets used for reference checking in Experimental Oberon). Thus, if the Oberon-2 language needs to be supported in Experimental Oberon, procedure variable offsets *can* also be *appended* to the existing fields of each type descriptor. This was realized in an earlier version of Experimental Oberon. Reverting back to that variant would require only a one-line change to the compiler (procedure `ORG.BuildTD`) and to the scan phase of reference checking (procedure `Kernel.Scan`). Alternatively, one could also choose to adapt the Oberon-2 implementation, such that the Oberon-2 method table is allocated somewhere else (e.g., in the heap at module load time – as is realized in the original Object Oberon implementation for example).

The compiler generating the modified type descriptors for dynamic objects and the descriptors for global module data, the format of the Oberon object file containing them and the module loader transferring them from object file into memory have been adjusted accordingly¹⁷.

Although the operation of *reference checking*, as sketched above, may appear expensive at first, in practice, there is no performance issue. It is comparable in complexity to the process of garbage collection and thus barely noticeable. In addition, unloading of modules is usually rare (except, perhaps, during development, where however the number of references to a module tends to be small), while modules that manage data structures shared by client modules (such as a viewer manager) are often never unloaded at all. Thus, the presented solution, which was mainly chosen for its simplicity, appears to be amply sufficient for most practical purposes.

Its main shortcoming then is that it requires *additional* run-time information to be present in type descriptors *solely* for the purpose of reference checking, which in turn is *only* needed in the relatively infrequent case of module releases. However, since there exists only one type descriptor per declared record *type* rather than one per allocated heap *object*, the additional memory requirements are negligible¹⁸.

An additional downside is that when a module cannot be removed from memory due to pre-existing references, the user usually does not know why the removal has failed. However, since in this case the module is removed only from the *list* of loaded modules and not from *memory*, we don't consider this issue as serious. In addition, this shortcoming can be easily alleviated by displaying the *names* of the modules containing the references¹⁹.

5. System building tools

A minimal version of the Oberon system *building tools* has been added, consisting of the two modules *Linker*²⁰ and *Builder*. They provide the necessary tools to establish the prerequisites for the *regular* Oberon startup process²¹.

When the power to a computer is turned on or the reset button is pressed, the computer's *boot firmware* is activated. The boot firmware is a small standalone program permanently resident in the computer's read-only store, such as a read-only memory (ROM) or a field-programmable read-only memory (PROM), which is part of the computer's hardware.

In Oberon, the boot firmware is called the *boot loader*, as its primary task is to *load* a valid *boot file* (a pre-linked binary containing a set of compiled Oberon modules) from a valid *boot source* into memory and then transfer control to its *top* module (the module that directly or indirectly imports all other modules in the boot file). Then its job is done until the next time the computer

¹⁷ See procedures *ORG.BuildTD*, *ORG.Close* and *Modules.Load*.

¹⁸ Adding procedure variable offsets to type descriptors is, strictly speaking, not necessary, as the compiler can always rearrange the memory layout of a record such that all procedure variable fields are placed in a contiguous section at the beginning of it, thereby making their offsets implicit. Even the number of procedure variable fields could be encoded in one of the existing fields of its associated type descriptor, for example in the sentinel at the end of the pointer offset section. Thus, it is possible to realize the reference checking scheme without additional memory requirements in type descriptors. However, we refrain from implementing this refinement for two main reasons. First, it would increase the complexity of the compiler. At first glance, it might seem that in order to rearrange a record's field list, only a small change needs to be made to a single procedure in the compiler (*ORG.RecordType*). However, a complication arises because records may recursively contain other records, each again containing procedure variables at arbitrary offsets. While it is always possible to flatten such recursive record structures, it would make other record operations more complex. For example, assignments to subrecords would become less natural, because their fields would no longer be placed in a contiguous section in memory. Thus, we would no longer obtain a simple and uniform implementation covering both dynamic records and global module data. Second, the memory savings in the module areas holding type descriptors would be marginal, given that there exists only one descriptor per *declared* record type rather than one per *allocated* heap object. In addition, we believe that most applications should be programmed in the conventional programming style. Therefore, installed procedures should be rare, while in the few places where they do exist, there are typically only a few of them. For example, in the Oberon system, the object-oriented programming style is restricted to the viewer system, which provides distributed control in the form of installed handler procedures (of which there is typically only one per viewer type). On balance, the benefit obtained by saving a few fields in a small number of type descriptors appears negligible and therefore the additional complexity required to implement this refinement would be hard to justify.

¹⁹ In the case of dynamic records, one can display the name of the module declaring its *type*, which is not necessarily the module rooting the data structure holding the record. One could however modify the reference checking algorithm such that each module is checked individually for dynamic references, in which case also the data structure roots are known.

²⁰ The linker has been included from a different source (<https://github.com/charlesap/lo>). It was adapted for Experimental Oberon's object file format (which includes a modified module descriptor, modified type descriptors for dynamically allocated records and an additional section in the module block holding global procedure variable references).

²¹ Currently not implemented is a tool to prepare a disk initially – which consists of a single *'Kernel.PutSector'* statement that initializes the root page of the file directory (sector 1).

is restarted or the reset button is pressed. In general, there is no need to modify the boot loader (*BootLoad.Mod*). If one really has to, one typically has to resort to proprietary tools to load the boot loader onto the specific hardware platform used.

There are currently two valid boot sources in Oberon: a local disk, realized using a Secure Digital (SD) card in Oberon 2013, and a communication channel, realized using an RS-232 serial line. The *default* boot source is the local disk. It is used by the *regular* Oberon startup process each time the computer is powered on or the reset button is pressed.

The command *Linker.Link* links a set of Oberon object files together and generates a valid boot file from them. The linker is almost identical to the regular module loader (*Modules.Load*), except that it writes the result to a file instead of loading and linking the modules in memory.

The command *Builder.Load* loads a valid boot file, as generated by *Linker.Link*, onto the *boot area* (sectors 2-63 in Oberon 2013) of the local disk, one of the two valid boot sources.

The format of the boot file is *defined* to exactly mirror the standard Oberon storage layout²². In particular, location 0 in the boot file (and later in memory once it has been loaded by the boot loader) contains a branch instruction to the initialization sequence of the *top* module of the boot file. Thus, the boot loader can simply transfer the boot file byte for byte from a valid boot source into memory and then branch to location 0 – which is precisely what it does²³.

In sum, to generate a new regular Oberon boot file and load it onto the local disk's boot area, one can execute the following commands *on* the system that is to be modified:

```
ORP.Compile Kernel.Mod FileDir.Mod Files.Mod Modules.Mod ~ ... compile the modules of the inner core
Linker.Link Modules ~ ... create a regular boot file (Modules.bin)
Builder.Load Modules ~ ... load boot file onto the disk's boot area
```

Note that the last command overwrites the disk's boot area of the *running* system. A backup of the disk is therefore recommended before experimenting with new *inner cores*²⁴.

When *adding* new modules to a boot file, the need to call their module initialization bodies during stage 1 of the boot process may arise, i.e. when the boot file is loaded into memory by the boot loader during system restart or when the reset button is pressed. We recall that the boot loader merely *transfers* the boot file byte for byte from a valid boot source into memory, but does not call the module initialization sequences of the transferred modules (this is, in fact, why the *inner core* modules *Kernel*, *FileDir* and *Files* don't *have* module initialization bodies – they wouldn't be executed anyway).

The easiest way to add a new module *with* a module initialization body to a boot file is to move its initialization code to an exported procedure *Init* and call it from the top module of the modules in the boot file. This is the approach chosen in Original Oberon, which uses module *Modules* as the top module of the *inner core*.

An alternative solution is to extract the starting addresses of the initialization bodies of the just loaded modules from their module descriptors in main memory and simply call them, as shown in procedure *InitMod*²⁵ below (see chapter 6 of the book *Project Oberon* for a detailed

²² See chapter 8.1, page 103, of the book *Project Oberon (2013 Edition)* for a detailed description of Oberon's storage layout.

²³ After transferring the boot file, the boot loader also deposits some additional key data in fixed memory locations, to allow proper continuation of the boot process.

²⁴ When using an Oberon emulator (<https://github.com/pdewacht/oberon-risc-emu>) on a host system, one can simply make a copy of the directory containing the disk image.

²⁵ Procedure *InitMod* could be placed in modules *Oberon* or *Modules* (note: the data structure rooted in the global variable *Modules.root* is transferred as part of the boot file).

description of the format of a *module descriptor* in memory; here it suffices to know that it contains a pointer to a list of *entries* for exported entities, the first one of which points to the initialization code of the module itself).

```

PROCEDURE InitMod(name: ARRAY OF CHAR);
1  VAR mod: Modules.Module; body: Modules.Command; w: INTEGER;
2  BEGIN mod := Modules.root;
3    WHILE (mod # NIL) & (name # mod.name) DO mod := mod.next END ;
4    IF mod # NIL THEN SYSTEM.GET(mod.ent, w);
5      body := SYSTEM.VAL(Modules.Command, mod.code + w); body
6    END
  END InitMod;

```

In the following example, module *Oberon* is chosen as the new top module of the *inner core*, while module *System* is configured to be the new top module of the *outer core*.

```

MODULE Modules;                                ... old top module of the inner core, now just a regular module
1  IMPORT SYSTEM, Files;
2  ...
3  BEGIN Init                                    ... no longer loads module Oberon (as in Original Oberon)
  END Modules.

MODULE Oberon;                                  ... new top module of the inner core, now part of the boot file
1  ...
2  BEGIN                                        ... boot loader will branch to here after transferring the boot file
3    InitMod(„Modules“);                      ... must be called first (establishes a working file system)
4    InitMod(„Input“);
5    InitMod(„Display“);
6    InitMod(„Viewers“);
7    InitMod(„Fonts“);
8    InitMod(„Texts“);
9    ...
10   Modules.Load(„System“, Mod);              ... load the outer core using the regular Oberon loader
11   Loop                                      ... transfer control to the Oberon central loop
  END Oberon.

```

To build a modified *inner core* for this new module configuration and load it onto the local disk's *boot area*, one can execute the following commands:

```

ORP.Compile Kernel.Mod FileDir.Mod Files.Mod Modules.Mod ~
ORP.Compile Input.Mod Display.Mod Viewers.Mod ~
ORP.Compile Fonts.Mod Texts.Mod Oberon.Mod ~      ... compile the modules of the modified inner core
Linker.Link Oberon ~                             ... create a new regular boot file (Oberon.bin)
Builder.Load Oberon ~                             ... load the boot file onto the local disk's boot area

```

This module configuration reduces the number of stages in the regular Oberon *boot process* from 3 to 2, thereby streamlining it somewhat, at the expense of extending the *inner core*. If one prefers to keep the *inner core* minimal, one could also choose to extend the *outer core* instead, e.g. by including module *System* and all its imports. This in turn would have the disadvantage that the viewer complex and the system tools are “locked” into the *outer core*. However, an Oberon system without a viewer manager hardly makes sense, even in closed server environments. As an advantage, we note that such an extended *outer core* can more easily be replaced on the fly (by unloading and reloading a suitable *subset* of modules).

* * *