

Enhanced FPGA Oberon-07 Compiler

Andreas Pirklbauer

12.12.2018

The official FPGA Oberon-07 compiler¹ has been enhanced with various new features, including

- Type-bound procedures (Oberon-2 style)²
- Dynamic heap allocation procedure for fixed-length and open arrays (Oberon-2 style)
- Numeric case statement
- Exporting and importing of string constants
- Forward references and forward declarations of procedures
- No access to intermediate objects within nested scopes
- Module contexts

The combined *total* implementation cost in the compiler of these new features in source lines of code (*sloc*)³, broken down by compiler module, is shown below:

Compiler module	FPGA Oberon	Experimental Oberon	Difference	Percent
ORS (scanner)	293	293	0	0 %
ORB (base)	394	440	46	+ 11.7 %
ORG (generator)	984	1112	128	+ 13.0 %
ORP (parser)	949	1143	194	+ 20.4 %
Total	2620	2988	368	+ 14.0 %

The detailed implementation cost of the various new features is as follows⁴:

Feature	Source lines of code
Type-bound procedures	~220
Dynamic heap allocation procedure for fixed-length and open arrays	~30
Numeric case statement	~80
All other features combined	~40
Total	~370

Type-bound procedures (Oberon-2 style)

Globally declared procedures may be associated with a record type declared in the same module. The procedures are said to be *bound* to the record type. The binding is expressed by the type of the *receiver* in the heading of a procedure declaration. The receiver may be either a variable parameter of record type *T* or a value parameter of type POINTER TO *T* (where *T* is a record type). The procedure is bound to the type *T* and is considered local to it.

¹ <http://www.projectoberon.com>

² Mössenböck H., Wirth N.: *The Programming Language Oberon-2. Structured Programming*, 12(4):179-195, 1991

³ Not counting empty lines

⁴ Not counting approximately 100-150 additional lines of source code in modules Kernel, Modules and System to complement the implementations of type-bound procedures, the dynamic heap allocation procedure NEW(p, len) and module contexts.

ProcedureHeading = PROCEDURE [Receiver] IdentDef [FormalParameters].
 Receiver = "(" [VAR] ident ":" ident ")".

If a procedure P is bound to a type T_0 , it is implicitly also bound to any type T_1 which is an extension of T_0 . However, a procedure P' (with the same name as P) may be explicitly bound to T_1 in which case it overrides the binding of P . P' is considered a *redefinition* of P for T_1 . The formal parameters of P and P' must match. If P and T_1 are exported, P' must be exported too.

If v is a designator and P is a type-bound procedure, then $v.P$ denotes that procedure P which is bound to the dynamic type of v . This may be a different procedure than the one bound to the static type of v . v is passed to P 's receiver according to the standard parameter passing rules.

If r is a receiver parameter declared with type T , $r.P^\wedge$ (pronounced *r.P-referenced*) denotes the (redefined) procedure P bound to the base type of T .

In a forward declaration of a type-bound procedure the receiver parameter must be of the *same* type as in the actual procedure declaration. The formal parameter lists of both declarations must be identical.

```

1  TYPE Tree = POINTER TO Node;
2  Node = RECORD key : INTEGER;
3    left, right: Tree
4  END ;
5
6  CenterTree = POINTER TO CenterNode;
7  CenterNode = RECORD (Node) width: INTEGER;
8    subnode: Tree
9  END ;
10
11 PROCEDURE (t: Tree) Insert (node: Tree);
12   VAR p, father: Tree;
13 BEGIN p := t;
14   REPEAT father := p;
15     IF node.key < p.key THEN p := p.left
16     ELSIF node.key > p.key THEN p := p.right
17     ELSE p := NIL
18   END
19 UNTIL p = NIL;
20 IF node.key < father.key THEN father.left := node ELSE father.right := node END;
21 node.left := NIL; node.right := NIL
22 END Insert;
23
24
25 PROCEDURE (t: CenterTree) Insert (node: Tree); (*redefinition*)
26 BEGIN Out.Int(node(CenterTree).width, 3);
27   t.Insert^(node) (*calls the Insert procedure bound to Tree*)
28 END Insert;
```

Dynamic heap allocation procedure for fixed-length and open arrays (Oberon-2 style)

If p is a variable of type $P = \text{POINTER TO } T$, a call of the predefined procedure *NEW* allocates a variable of type T in free storage at run time. The type T can be a record type or an array type.

If T is a record type or an array type with *fixed* length, the allocation has to be done with

NEW(p)

If T is an *open* array type, the allocation has to be done with

NEW(p , len)

where T is allocated with the length given by the expression len , which must be an integer type.

In either case, a pointer to the allocated variable is assigned to p . This pointer p is of type P , while the referenced variable p^{\wedge} (pronounced *p-referenced*) is of type T .

If T is a record type, a field f of an allocated record p^{\wedge} can be accessed as $p^{\wedge}.f$ or as $p.f$. If T is an array type, the elements of an allocated array p^{\wedge} can be accessed as $p^{\wedge}[0]$ to $p^{\wedge}[len-1]$ or as $p[0]$ to $p[len-1]$, i.e. record and array selectors imply dereferencing.

If T is an array type, its element type can be a *record*, *pointer*, *procedure* or a *basic* type (BYTE, BOOLEAN, CHAR, INTEGER, REAL, SET), but not an *array* type (no multi-dimensional arrays).

Example:

```
1  MODULE Test;
2    TYPE R = RECORD x, y: INTEGER END ;
3
4    A = ARRAY OF R;           (*open array*)
5    B = ARRAY 20 OF INTEGER;  (*fixed-length array*)
6
7    P = POINTER TO A;         (*pointer to open array*)
8    Q = POINTER TO B;         (*pointer to fixed-length array*)
9
10   VAR a: P;
11       b: Q;
12
13   PROCEDURE New1*;
14   BEGIN NEW(a, 100); a[53].x := 1
15   END New1;
16
17   PROCEDURE New2*;
18   BEGIN NEW(b); b[3] := 2
19   END New2;
20
21
22  END Test.
```

If the variable passed as parameter p of $NEW(p)$ or $NEW(p, len)$ is a *named* global pointer variable pointing to a record or an array type, the object referenced by p will be marked during the *mark* phase of the garbage collector and collected during the *scan* phase, if it is no longer referenced, i.e. both record *and* array blocks are garbage-collected.

The following rules and restrictions apply:

- Bounds checks on *fixed-length* arrays are performed at *compile* time.
- Bounds checks on *open* arrays are performed at *run* time.
- If P is of type $P = \text{POINTER TO } T$, the type T must be a *named* record or array type⁵.

⁵ Restricting pointers to *named* arrays is consistent with the official Oberon-07 compiler, which only allows pointers to *named* records.

Allocating dynamic arrays requires a modified version of module *Kernel*, which introduces a new *kind* of heap block (*arrayblk* in addition to *recordblk*)⁶. The implementation of garbage collection on open arrays is similar to implementations in earlier versions of the Original Oberon system⁷.

Numeric case statement

The official Oberon-07 language report⁸ allows numeric CASE statements, which are however not implemented in the official release⁹. The modified Oberon-07 compiler brings the compiler in line with the language report, i.e. it also allows *numeric* CASE statements (*CASE integer OF*, *CASE char OF*) in addition to *type* CASE statements (*CASE pointer OF*, *CASE record OF*)¹⁰.

Implemented syntax:

```
CaseStatement = CASE expression OF case {"|" case} [ELSE StatementSequence] END.
case          = CaseLabelList ":" StatementSequence.
CaseLabelList = LabelRange {"|" LabelRange}.
LabelRange   = label [".." label].
label        = integer | string | qualident.
```

Example:

```
1 CASE i OF
2     2..5: k := 1           (*lower case label limit = 2*)
3     | 8..10: k := 2
4     | 13..15: k := 3
5     | 28..30, 18..22: k := 4
6     | 33..36, 24: k := 5   (*higher case label limit = 36*)
7 END
```

The essential property of the *numeric* CASE statement is that it represents a *single*, indexed branch, which selects a statement sequence from a set of cases according to an index value. This is in contrast to a cascaded conditional statement which contains multiple branches. Case statements are recommended only if the set of selectable statements is reasonably large.

The compiler constructs a *jump table* of branch statements (containing the branch distances as operands) to the various component statements, leading to a *constant* number of instructions needed for any selection in a numeric CASE statement. The selection ("switch") is generated by procedure *ORG.CaseHead*, using a jump table generated by procedure *ORG.CaseTail*.

Jump tables are located in the *code* section of a module and are addressed relative to the *static base* (SB) of the module. The offset to the static base is restricted to a maximum of 64KB (one could also have placed a table of jump table offsets in the module's area for constants, which is sometimes used if the processor features an indexed branch instruction, as is typically the case in CISC processors, but not in the RISC processor used by FPGA Oberon).

The following rules and restrictions apply:

⁶ A call to *NEW(p, n)*, where *p* is a pointer to an array of *BYTE*, is equivalent to a call to the low-level procedure *SYSTEM.NEW(p, n)* provided in some implementations of the Original Oberon system to allocate a storage block of *n* bytes ("sysblk"). In our implementation, no such additional low-level procedure exported by the pseudo-module *SYSTEM* is necessary.

⁷ See, for example, "Oberon Technical Notes: Garbage collection on open arrays", J. Tempel, ETH technical report, March 1991.

⁸ <http://www.inf.ethz.ch/personal/wirth/Oberon/Oberon07.Report.pdf> (as of 3.5.2016)

⁹ <http://www.projectoberon.com>

¹⁰ <http://github.com/andreaspirklbauer/Oberon-numeric-case-statement>

- Case labels must have values between 0 and 255. This makes them ideally suitable for use with CHAR selectors, without imposing any serious restrictions on integer selectors.
- Case expression values that are between 0 and the highest case label limit, but do not correspond to any case label value in the source text, do *not* cause an error termination, i.e. we treat such events as "empty" actions, unless an ELSE clause is present.
- Case expression values that are *higher* than the highest case label limit *do* cause an error termination (trap 1), regardless of whether an ELSE clause is present. This additional rule has been introduced in order to keep the size of the generated jump table reasonably small, which helps optimize the common case where fewer than 256 entries are actually needed.
- Case statements do *not* allow for the empty case, unlike in the official Oberon-07 language, where a case is defined using the following grammatical production *with* brackets [and]

case = [CaseLabelList ":" StatementSequence].

The *inclusion* of the empty case would allow the insertion of superfluous bars similar to the insertion of superfluous semicolons between statements, which we view as .. superfluous ☺

The ELSE clause is implemented for both the *numeric* and the *type* CASE statement, even though it is not part of the official Oberon-07 language definition. If the clause is present, the statement sequence following the ELSE keyword is executed if the value of the case expression does *not* correspond to any case label value in the source text *and* (in the numeric case) lies between 0 and the highest case label limit.

At first glance, this may seem to be somewhat in contradiction with conventional wisdom in programming language design, which suggests that the ELSE clause in a language construct should normally be reserved for the *exceptional* cases only, i.e. those that are neither numerous among the possible cases in the source text nor do occur frequently at run time. The presence of an ELSE clause in the source text may also obfuscate the thinking of the programmer, for example if program execution falls through to the ELSE clause unintentionally. In general, language constructs that allow program execution to continue or “fall through” to the next case or a “default” case are not recommended.

We have nevertheless opted to re-introduce the ELSE clause for the following reason. In our implementation, case expression values *higher* than the highest case label limit cause an error termination. This additional rule effectively limits the *total* range of possible case label values, making it easier to align it with the actual use case. Consequently, if a CASE statement is well designed, the ELSE clause is much more likely to *actually* be used for the exceptional cases – as it should. In addition, there exist genuine examples where an ELSE clause appears useful.

Exporting and importing of string constants

The official Oberon-07 language report allows exporting and importing of string constants, but the compiler does not support it. The modified Oberon-07 compiler implements this feature¹¹.

Example:

```

1  MODULE M;
2    CONST s* = "This is a sample string";      (*exported string constant*)
3  END M.
4
```

¹¹ <http://github.com/andreaspirklbauer/Oberon-importing-string-constants>

```

5  MODULE N;
6    IMPORT Texts, Oberon, M;
7    VAR W: Texts.Writer;
8
9    PROCEDURE P*;
10   BEGIN Texts.WriteString(W, M.s); Texts.WriteLn(W); Texts.Append(Oberon.Log, W.buf)
11   END P;
12
13  BEGIN Texts.OpenWriter(W)
14  END N.

```

Exported *string constants* are treated like (pre-initialized) exported *variables*. The symbol file contains the string's *export number* and *length*, but not the string itself. The object file contains the actual string, together with its *location* in the exporting module's area for *data*.

Forward references and forward declarations of procedures

The modified Oberon compiler implements forward references of procedures for 2 use cases¹²:

Use case A: To make references among nested procedures more efficient:

If a procedure Q which is local to another procedure P refers to the enclosing procedure P, as in

```

1  PROCEDURE P;
2    PROCEDURE Q;
3      BEGIN (*body of Q*) P          (*forward reference from Q to P, as the body of P is not compiled yet *)
4    END Q;
5  BEGIN (*body of P*) ...
6  END P;

```

then the official Oberon-07 compiler generates the following code:

```

20  P'   BL 10          ... forward branch to line 31 (the body of P)
21  Q    body of Q
...
31  P    body of P

```

... any calls from Q to P are BACKWARD jumps to line 20 and from there forward to line 31

whereas the modified compiler generates the following, more efficient, code:

```

20  Q    body of Q
...
30  P    body of P

```

... any calls from Q to P are FORWARD jumps to line 30, fixed up when P is compiled

i.e. it does not generate an extra forward jump in line 20 around Q to the body of P and backward jumps from Q to line 20. With the official compiler, the extra BL instruction in line 20 is generated, so that Q can call P (as the body of Q is compiled before the body of P).

Use case B: To implement forward declarations of procedures:

Forward declarations of procedures have been eliminated in Oberon-07, as they can always be eliminated from any program by an appropriate nesting or by introducing procedure variables¹³.

¹² <http://github.com/andreaspirklbauer/Oberon-forward-references-of-procedures>

¹³ See section 2 of www.inf.ethz.ch/personal/wirth/Oberon/PortingOberon.pdf

Whether forward declarations of procedures *should* be re-introduced into the Oberon language, can of course be debated. Here, we have re-introduced them for three main reasons:

- Direct procedure calls are more efficient than using procedure variables.
- Legacy programs that contain forward references of procedures are now accepted again.
- Introducing forward declarations of procedures added only about 10 lines of source code.

Forward declarations of procedures are implemented in exactly the same way as in the original implementation before the Oberon-07 language revision, i.e.,

- They are explicitly specified by `^` following the symbol `PROCEDURE` in the source text.
- The compiler processes the heading in the normal way, assuming its body to be missing. The newly generated object in the symbol table is marked as a forward declaration.
- When later in the source text the full declaration is encountered, the symbol table is first searched. If the given identifier is found and denotes a procedure, the full declaration is associated with the already existing entry in the symbol table and the parameter lists are compared. Otherwise a multiple definition of the same identifier is present.

Note that our implementation *both* global and local procedures can be declared forward.

Example:

```
1  MODULE M;
2  PROCEDURE^ P(x, y: INTEGER; z: REAL);           (*forward declaration of P*)
3
4  PROCEDURE Q*;
5  BEGIN P(1, 2, 3.0)                             (*Q calls P which is declared forward*)
6  END Q;
7
8  PROCEDURE P(x, y: INTEGER; z: REAL);           (*procedure body of P*)
9  BEGIN ...
10 END P;
11
12 END M.
```

No access to intermediate objects within nested scopes

The official Oberon-07 language report disallows access to *all* intermediate objects from within nested scopes. The modified compiler brings the compiler in line with the language report, i.e. it also disallows access to intermediate *constants* and *types* within nested scopes, not just access to intermediate *variables*¹⁴.

Like the official compiler, the modified compiler adopts the convention to implement *shadowing through scope*. This means when two objects share the same name, the one declared at the narrower scope hides, or shadows, the one declared at the wider scope. In such a situation, the shadowed element is not available in the narrower scope.

The official Oberon-07 compiler already issues an error message, if intermediate *variables* are accessed within nested scopes, *regardless* of whether a global variable with the same name exists or not (e.g. in line 23 of the program below). With the modified compiler, the same error message is now *also* issued for intermediate *constants* (line 19) and *types* (lines 14, 16).

¹⁴ <http://github.com/andreaspirklbauer/Oberon-no-access-to-intermediate-objects>

Example:

```
1  MODULE Test;
2  CONST C = 10;           (*global constant C - shadowed in Q*)
3  TYPE G = REAL;         (*global type G - NOT shadowed in Q*)
4  T = REAL;              (*global type T - shadowed in Q*)
5  VAR A,                  (*global variable A - NOT shadowed in Q*)
6  B: INTEGER;            (*global variable B - shadowed in Q*)
7
8  PROCEDURE P;           (*global procedure P*)
9  PROCEDURE Q;          (*intermediate procedure Q*)
10  CONST C = 20;         (*intermediate constant C - shadows global constant C*)
11  TYPE T = INTEGER;     (*intermediate type T - shadows global type T*)
12  VAR B: INTEGER;       (*intermediate variable B - shadows global variable B*)
13
14  PROCEDURE R(x: T): T;  (*access to int. type T allowed in original, NOT allowed in modified compiler*)
15  VAR i: INTEGER;
16  q: T;                 (*access to int. type T allowed in original, NOT allowed in modified compiler*)
17  g: G;                 (*access to global type G (not shadowed) allowed in BOTH compilers*)
18  BEGIN (*R*)
19  i := C;               (*access to int. constant C allowed in original, NOT allowed in modified compiler*)
20  P;                   (*access to global procedure P allowed in BOTH compilers*)
21  Q;                   (*access to intermediate procedure Q allowed in BOTH compilers*)
22  i := A;               (*access to global variable A (not shadowed) allowed in BOTH compilers*)
23  i := B;               (*access to intermediate variable B NOT allowed in either compiler*)
24  RETURN i
25  END R;
26  END Q;
27  END P;
28
29  END Test.
```

Module contexts

The modified Oberon-07 compiler introduces *module contexts*, originally introduced for the A2 operating system¹⁵. A module context acts as a single-level name space for modules. It allows modules with the same name to co-exist within different contexts. The syntax is defined as:

```
Module    = MODULE ident [IN ident] “;”
Import    = IMPORT ident [“:=” ident ] [IN ident] “;”
```

In the first line, the optional identifier specifies the name of the context the module belongs to. In the second line, it tells the compiler in which context to look for when importing modules.

Module contexts are implemented as follows:

- Module contexts are specified within the *source* text of a module, as an optional feature. If a context is specified, the name of the source file itself typically (but not necessarily) contains a prefix indicating its module context, for example *Oberon.Texts.Mod* or *EO.Texts.Mod*.
- If a module context is specified in the source text, the compiler will automatically generate the output files *contextname.modulename.smb* and *contextname.modulename.rsc*, i.e. the module contexts of symbol and object files is encoded in their file names.

¹⁵ <http://www.ocp.inf.ethz.ch/wiki/Documentation/Language?action=download&upname=contexts.pdf>

- If no module context is specified in the source text, the output files *modulename.rsc* and *modulename.smb* are generated.
- In Experimental Oberon, the module context "EO" is implicitly specified at run time. Thus, the module loader will first look for the file *EO.modulename.rsc*, then for *modulename.rsc*.
- A module belonging to a context can only import modules belonging to the same context, and vice versa (implementation restriction).

Example:

```

1  MODULE Test1 IN EO; (*Experimental Oberon*)
2    IMPORT Texts, Oberon;
3    VAR W: Texts.Writer;
4
5    PROCEDURE Go1*;
6    BEGIN Texts.WriteString(W, "Hello from module Test1 in context EO (Experimental Oberon)");
7      Texts.WriteLine(W); Texts.Append(Oberon.Log, W.buf)
8    END Go1;
9
10 BEGIN Texts.OpenWriter(W)
11 END Test1.
12
13 MODULE Test2 IN EO; (*Experimental Oberon*)
14   IMPORT Texts, Oberon, Test1 IN EO;
15   VAR W: Texts.Writer;
16
17   PROCEDURE Go2*;
18   BEGIN Texts.WriteString(W, "Hello from module Test2 in context EO (Experimental Oberon)");
19     Texts.WriteLine(W); Texts.Append(Oberon.Log, W.buf)
20   END Go2;
21
22 BEGIN Texts.OpenWriter(W)
23 END Test2.
```

* * *